PSpice Simulation of a Resistive Bridge

Use PSpice to plot v versus i for this resistive bridge over the range $-1\text{ A} \le i \le 1\text{ A}$.
Time permitting, add plots of v_2 and v_{23}.



Simulation Diagram



CIRCUIT FILE

```
PSpice Simulation of Resistive Bridge
i   0 1  dc  1
*  Note polarity per passive convention
R12 1 2  6
R13 1 3  2
R23 2 3  12
R20 2 0  3
R30 3 0  12
.dc lin i -1, 1, 0.1; sweeps i over -1 to 1 in 0.1-A steps
.probe
* v = v(1)
.end
```

$v = V(1), R_{eq} = \Delta v/\Delta i = 5\ \Omega, v_2 = V(2), v_{23} = V(2,3) = V(2) - V(3)$ (has negative slope)

PSnice Simulation of the v-i Curve of a Source Network
(Example 2.12)

The circuit file below simulates the source-load circuit in Fig. 2.27a. Run the Probe file to get the *v-i* curve as R_L sweeps from 0.5 Ω to 50 Ω.

The horizontal axis can be changed to current i = i(v35) by selecting **Axis Variable** under **Plot + X Axis Settings**. The range may be set by the user as −1.2 A to 0 A, and cursors under **Tools** may be used to determine the coordinates of the end points. The plot should then look like Fig. 2.27b. The Thévenin parameters of the source network can now be determined as in Example 2.12.

CIRCUIT FILE

```
Circuit file for Example 2.12
vin  1 0  dc 1
R12  1 2  250
R20  2 0  4k
R23  2 3  2k
R34  3 4  80
E04  0 4  (2,0) 100; control voltage is vx = v(2)
v35  3 5  dc 0; senses i = i(v35)
RL   5 0  {RL}; variable resistance
.param RL=.5; nominal value
.dc lin param RL .5, 50, .5; sweeps RL from 0.5 to 50 ohms
.probe
.end
```

PSpice Simulation of Power Transfer

The circuit file below sweeps the load resistance over $0.1\ \text{k}\Omega \le R_L \le 16\ \text{k}\Omega$. Run Probe to obtain a plot showing the power p delivered to R_L. Then use the cursor under **Tools** to find the maximum value of p and the corresponding value of R_L. (*Hint*: The cursor menu has a "max" command.) Time permitting, add a plot of the internal power dissipation p_s. (*Hint*: The square of any variable x can be obtained in Probe by writing $x*x$.)



CIRCUIT FILE

```
PSpice Simulation of Power Transfer
v10 1 0  dc  24
R12 1 2  6k
R20 2 0  30k
v23 2 3  dc  0; senses i = i(v23)
R30 3 0  {RL}; variable resistance
.param RL = 0.1k; nominal value
.dc lin param RL 0.1k, 16k, 0.1k; sweeps RL
.probe
.end
```

p = V(3)*I(v23), p_{max} = 20 mW when R_L = 5 kΩ
p_s = V(1,2)*V(1,2)/6k+V(2)*V(2)/30k (This curve differs from Fig. 3.6 because p_s is an internal quantity and the actual circuit is not like Fig. 3.3a.)

PSpice Simulation of a Charging Capacitor

The circuit file below simulates a 20-nF capacitor charging to 100 V through resistance R. The capacitor has an initial voltage of 0 V established by `ic = 0`. The `.param` and `.step` statements carry out the simulation for $R = 250\ \text{k}\Omega, 500\ \text{k}\Omega$, and $1\ \text{M}\Omega$.

Run Probe, selecting all cases, to obtain a multi-trace plot of the capacitor voltage V(2) versus time. Then use the cursor to estimate the time t_1 when the voltage has risen to 99 V. Put your results in the following table, along with the calculated value of t_1/RC

R	250 kΩ	500 kΩ	1 MΩ
RC	5 ms	10 ms	20 ms
t_1	**22.8 ms**	**46.0 ms**	**92.1 ms**
t_1/RC	**4.56**	**4.60**	**4.605**

CIRCUIT FILE

```
Charging capacitor
v  1  0  dc   100
R  1  2  {R}
C  2  0  20n  ic = 0; sets vC(0) = 0
.param  R = 250k
.step param  R  list  250k  500k  1meg
.tran  .1  .1  uic
.probe
.end
```

Conclusion: The charging transient is never over, but the voltage is essentially at its final value after about 5 × RC.

PSpice Simulation of Parallel Resonance, Including Coil Resistance



The circuit file below simulates this parallel-resonant circuit with two values of the coil resistance, $R_w = 10^{-9}\Omega \approx 0$ and $R_w = 2\ \Omega$. When $R_w = 0$, the circuit has $Q_{par} \approx 10$ and $f_0 = \omega_0/2\pi \approx 8$ kHz.

Run the Probe file and proceed as follows:

(a) Select both cases
(b) Set linear X-axis
(c) Plot vm(1)@1 and vp(1)@1. Note: $\underline{Z} = \underline{V}/\underline{I} = $ v(1), so $|Z| = $ vm(1) and $\angle Z = $ vp(1).
(d) Plot im(C)@1 and im(L)@1
(e) Plot vm(1)@1 and vm(1)@2
Expand 0 to 1 kHz
Expand 6 kHz to 10 kHz, and use cursors to measure the peaks (value, location) of $|Z|$

Based on this simulation, what do you conclude about the affects of coil resistance in parallel resonance?

CIRCUIT FILE

```
Parallel Resonance with Coil Resistance
I  0 1  ac 1 ; AC current source with im = 1 A, ip = 0 deg
R  1 0  100
C  1 0  2u
L  1 2  .2m
Rw 2 0  {Rvar}
.param Rvar = 2
.step param Rvar list 1n, 2 ; Rw = 1E-9 (approximates 0), 2
* Linear frequency sweep, 400 points, 10 Hz to 16 kHz
.ac lin 400 10 16k
.probe
.end
```

PSpice Simulation of First-order Pulse Response

The circuit file below simulates a series RC circuit driven by a pulsed voltage. The pulse starts at $t = 0$ with 1-V height and lasts for duration $D = 1$ ms. The initially uncharged capacitor has value $C = 1\ \mu F$. The transient simulation is 3-ms long, and it steps the resistance value so that $R = 100\ \Omega$, $1\ k\Omega$, and $10\ k\Omega$.

Run the Probe file and display the input voltage V(1) and the capacitor voltage V(2) for all three values of R. Then display the resistance voltage V(1,2) for all three values of R. Comment on the resulting pulse response, compared with the capacitor voltage.

CIRCUIT FILE

```
1st-order pulse response
vin 1 0 pwl 1m,1 1.01m, 0; approximates 1-ms pulse
R   1 2 {R}
C   2 0 1u IC=0; sets vC(0)=0
.param R=.1k
.step param. R list .1k 1k 10k
.tran 3m 3m uic
.probe
.end
```

The results for V(1) and V(2) should look like Fig. 9.12, with $D = 1$ ms and $\tau = 0.1$ ms, 1 ms, and 10 ms.

PSpice Simulation of Second-order Pulse Response

The circuit file below simulates a series *LRC* circuit driven by a pulsed voltage. The pulse starts at $t = 0$ with 1-V height and lasts for duration $D = 0.1$ s. The inductor and capacitor have no initial energy storage, and the values are $L = 0.1$ H and $C = 1/640$ F. The transient simulation is 0.3-s long, and it steps the resistance value so that $R = 5\ \Omega$, $16\ \Omega$, and $34\ \Omega$. Thus, the circuit is like Example 9.12.

Run the Probe file and display the input voltage V(1) and the capacitor voltage V(3) for all three values of *R*. Then display the resistance voltage V(2,3) for all three values of *R*. Comment on the resulting pulse response of V(3) and V(2,3).

CIRCUIT FILE

```
2nd-order pulse response
vin 1 0 pwl .1,1 .101, 0; approximates 0.1-s pulse
L   1 2 .1 IC=0; sets iL(0) = 0
R   2 3 {R}
C   3 0 {1/640} IC=0; sets vC(0)=0
.param R=5
.step param R list 5, 16, 34
.tran .3 .3 uic
.probe
.end
```

PSpice Simulation of Stability



Analysis of this circuit reveals that $H(s) = \dfrac{\underline{V}_C}{\underline{V}} = \dfrac{-40(s - 10)}{s^2 + (R - 40)s + 400}$

so $\alpha = (R - 40)/2, \omega_0^2 = 400$, and $p_1, p_2 = -\alpha \pm \sqrt{\alpha^2 - 400}$ or $-\alpha \pm j\sqrt{400 - \alpha^2}$

This circuit is simulated by the PSpice file below. Run Probe to obtain a multi-trace display of the waveform $v_C(t)$ for cases (1), (2), and (3) below. Repeat for cases (3) and (4), and for cases (4) and (5).

(1) OVERDAMPED—$R = 144\ \Omega, \alpha = 52 > \omega_0, p_1, p_2 = -52 \pm 48 = -4, -100$
$v_C(t) = A_1e^{-4t} + A_2e^{-100t}$

(2) CRITICALLY DAMPED—$R = 80\ \Omega, \alpha = 20 = \omega_0, p_1, p_2 = -20$
$v_C(t) = A_1e^{-20t} + A_2te^{-20t}$

(3) UNDERDAMPED—$R = 44\ \Omega, \alpha = 2 < \omega_0, p_1, p_2 = -2 \pm j19.9$
$v_C(t) = Ae^{-2t}\cos(19.9t + \phi)$

(4) OSCILLATOR—$R = 40\ \Omega, \alpha = 0, p_1, p_2 = \pm j20$
$v_C(t) = A\cos(20t + \phi)$

(5) UNSTABLE—$R = 36\ \Omega, \alpha = -2 < 0, p_1, p_2 = +2 \pm j19.9$
$v_C(t) = Ae^{+2t}\cos(19.9t + \phi)$

CIRCUIT FILE

```
Stability analysis, Vc = v(3)
V   0  1  dc  0; v(t) = 0
R   1  2  {Rvar}
.param Rvar = 144
.step param Rvar list  144  80  44  40  36
L   2  3  1    ic = 0 ; i(0) = 0
G   3  0  (2,3)  .1
C   3  0  2.5m ic = 1 ; vC(0) = 1 V
.tran 1m 1 uic; transient analysis with initial conditions
.probe
.end
```

PSpice Simulation of a Bode Plot



The circuit file below generates a frequency-sweep simulation of this circuit covering the range $100 \leq \omega \leq 10^8$ rad/s with $\underline{I}_{in} = 1$ A for the two cases $R = 4\ \Omega$ and $4\ \text{k}\Omega$.

(a) Use Probe to obtain the Bode plots of $H(s) = \underline{I}_{out}/\underline{I}_{in}$, with a split-screen display of $g(\omega)$ and $\theta(\omega)$. (*Hints:* $\omega = 6.28$*Frequency and $H(s) = \underline{I}_{out}$ when $\underline{I}_{in} = 1$.)

(b) Taking the case when $R = 4\ \text{k}\Omega$, generate a separate dB plot and construct the asymptotes to estimate the values of the poles and zeros.

CIRCUIT FILE

```
Pspice Bode plot
Iin  0 1  ac 1
R10  1 0  80
R12  1 2  {5*R}
L20  2 0  10m
R13  1 3  {R}
L34  3 4   5m; Iout = i(L34)
C40  4 0   50n
.param R=4
.step param R list 4 4k
.ac dec 50 1 100meg
.probe
.end
```

(b) Zero at $\omega = 0$, poles at $-W_1 \approx -5{,}000$ and $-W_2 \approx -750{,}000$ r/s.

PSpice Simulation of a Butterworth Lowpass Filter



This circuit is a third-order Butterworth filter with $R = 7.2$ kΩ and $C = 2.2$ nF. (Note that the first stage is a first-order lowpass filter.) Let

$$H_1(j\omega) = \underline{V}_1/\underline{V}_{in} \qquad H_{12}(j\omega) = \underline{V}_2/\underline{V}_1 \qquad H_2(j\omega) = \underline{V}_2/\underline{V}_{in}$$

The circuit file below gives a frequency-sweep simulation with $\underline{V}_{in} = 1 \angle 0°$

(a) Use Probe to obtain a multi-trace plot showing the dB gain curves g_1 and g_2 versus ω on a log axis for $10^3 \leq \omega \leq 10^6$. (*Hints:* ω = 6.28*Frequency, and $|\underline{V}_{in}| = 1$ so g_1 = vdb(1), etc.) Comment on these curves, and use the cursor to estimate the values of $g_1(0)$ and $g_2(0)$.

(b) Use the cursor to find the values of ω at which the gains are down by 1 dB, 3 dB, and 20 dB compared to $g_1(0)$ and $g_2(0)$. Comment on your results.

(c) Time permitting, add g_{12} to your plot. Comment on this curve, and estimate $g_{12}(0)$.

CIRCUIT FILE

```
Butterworth LPF
Vin 5 0 ac 1
R56 5 6    7.2k
C60 6 0    2.2n
E10 1 0    (6,0) 2
R13 1 3    7.2k
C32 3 2    2.2n
R34 3 4    7.2k
C40 4 0    2.2n
E20 2 0  (4,0) 2
.ac lin 1000 1 50k
.probe
.end
```

(a) **g_2 is a more selective filter; $g_1(0) \approx 6$ dB, $g_2(0) \approx 12$ dB**

(b) **Both filters have $\omega_{co} \approx 63.3$k (10 kHz), but g_2 is flatter in the passband and has a steeper rolloff.**

(c) **g_{12} has a "hump" slightly below 10 kHz, coming from the pair of complex poles.**